>

MATH 392 -- Seminar in Computational Commutative Algebra

Pappus' Theorem -- April 17, 2019

Theorem. Let A, B, C and A', B', C' be two collinear triples of points in the plane, then the three points

$$P = AB' \cap A'B, \; Q = AC' \cap A'C, \; R = BC' \cap B'C$$

are collinear.

To translate this to polynomial equations, we introduce coordinates as follows. Let's make the line containing the points *A, B, C* the line $V(y)$. Then we can take *A', B'* arbitrary and add a hypothesis to make *C'* lie on the line containing *A'* and *B':*

$$A = (0, 0), \; B = (u_1, 0), C = (u_2, 0),$$

$$A' = (u_3, u_4), B' = (u_5, u_6), \; C' = (x_1, x_2)$$

The hypotheses are first, C' is collinear with A' **and** B':

> $h_1 := (x_1 - u_3)\cdot(x_2 - u_6) - (x_1 - u_5)\cdot(x_2 - u_4);$

$$h_1 := (x_1 - u_3)\,(x_2 - u_6) - (x_1 - u_5)\,(x_2 - u_4) \tag{1}$$

Then we introduce the intersection points *P, Q, R* as above

$$P = (x_3, x_4), Q = (x_5, x_6), \; R = (x_7, x_8).$$

So $P \in AB'$ **and** $P \in A'B$:

> $h_2 := x_4 \cdot u_5 - x_3 \cdot u_6;$

$$h_2 := u_5\, x_4 - u_6\, x_3 \tag{2}$$

> $h_3 := (x_4 - u_4)\cdot(x_3 - u_1) - x_4\cdot(x_3 - u_3);$

$$h_3 := (x_4 - u_4)\,(x_3 - u_1) - x_4\,(x_3 - u_3) \tag{3}$$

Then $Q \in AC'$ **and** $Q \in A'C$:

> $h_4 := (x_6 - x_2)\cdot x_5 - (x_5 - x_1)\cdot x_6;$

$$h_4 := (x_6 - x_2)\, x_5 - (x_5 - x_1)\, x_6 \tag{4}$$

> $h_5 := (x_6 - u_4)\cdot(x_5 - u_2) - x_6\cdot(x_5 - u_3);$

$$h_5 := (x_6 - u_4)\,(x_5 - u_2) - x_6\,(x_5 - u_3) \tag{5}$$

Finally, $R \in BC'$ **and** $R \in B'C$:

> $h_6 := x_8\cdot(x_7 - x_1) - (x_8 - x_2)\cdot(x_7 - u_1);$

(6)

$$h_6 := x_8 (x_7 - x_1) - (x_8 - x_2)(x_7 - u_1) \quad \textbf{(6)}$$

```
> h[7] := x[8]·(x[7] - u[5]) - (x[8] - u[6])·(x[7] - u[2]);
```

$$h_7 := x_8 (x_7 - u_5) - (x_8 - u_6)(x_7 - u_2) \quad \textbf{(7)}$$

The conclusion is that *P, Q, R* are collinear

```
> g := (x[8] - x[6])·(x[7] - x[3]) - (x[8] - x[4])·(x[7] - x[5]);
```

$$g := (x_8 - x_6)(x_7 - x_3) - (x_8 - x_4)(x_7 - x_5) \quad \textbf{(8)}$$

```
> with(Groebner) :
> Htilde := [seq(h[i], i = 1..7), 1 - g·y];
```

$$Htilde := [(x_1 - u_3)(x_2 - u_6) - (x_1 - u_5)(x_2 - u_4), u_5 x_4 - u_6 x_3, (x_4 - u_4)(x_3 - u_1) - x_4 (x_3 - u_3), (x_6 - x_2) x_5 - (x_5 - x_1) x_6, (x_6 - u_4)(x_5 - u_2) - x_6 (x_5 - u_3), x_8 (x_7 - x_1) - (x_8 - x_2)(x_7 - u_1), x_8 (x_7 - u_5) - (x_8 - u_6)(x_7 - u_2), -((x_8 - x_6)(x_7 - x_3) - (x_8 - x_4)(x_7 - x_5)) y + 1] \quad \textbf{(9)}$$

```
> xvars := [seq(x[j], j = 1..8), y];
```

$$xvars := [x_1, x_2, x_3, x_4, x_5, x_6, x_7, x_8, y] \quad \textbf{(10)}$$

```
> Basis(Htilde, tdeg(op(xvars)));
```

$$[1] \quad \textbf{(11)}$$

Note the *u* - variables are omitted here in the specification of the grevlex order. Hence the conclusion *follows generically* from the hypotheses over C in this case. We have the same sort of behavior that we saw previously.

The conclusion is "almost" in the ideal generated by the hypotheses in the polynomial ring in both *x* - and *u* - variables, but there are "degenerate" configurations too introducing reducibility(!)

```
> H := [seq(h[i], i = 1..7)];
```

$$H := [(x_1 - u_3)(x_2 - u_6) - (x_1 - u_5)(x_2 - u_4), u_5 x_4 - u_6 x_3, (x_4 - u_4)(x_3 - u_1) - x_4 (x_3 - u_3), (x_6 - x_2) x_5 - (x_5 - x_1) x_6, (x_6 - u_4)(x_5 - u_2) - x_6 (x_5 - u_3), x_8 (x_7 - x_1) - (x_8 - x_2)(x_7 - u_1), x_8 (x_7 - u_5) - (x_8 - u_6)(x_7 - u_2)] \quad \textbf{(12)}$$

```
> HBasis := Basis(H, tdeg(seq(x[j], j = 1..8), seq(u[k], k = 1..6))) :
> nops(HBasis);
```

$$39 \quad \textbf{(13)}$$

```
>
```

Here is one of the Grobner basis, in factored form, and the conclusion:

```
> factor(HBasis[30]);
```

$$-u_4 u_6 (x_3 x_6 - x_3 x_8 - x_4 x_5 + x_4 x_7 + x_5 x_8 - x_6 x_7)(x_4 u_3 - u_4 x_3) \quad \textbf{(14)}$$

```
> expand(g);
```

$$x_3 x_6 - x_3 x_8 - x_4 x_5 + x_4 x_7 + x_5 x_8 - x_6 x_7 \quad \textbf{(15)}$$

```
> GenHBasis := Basis(H, tdeg(seq(x[j], j = 1..8)));
```

$$GenHBasis := [u_6\,x_7 + (u_2 - u_5)\,x_8 - u_2\,u_6,\ u_4\,x_5 + (u_2 - u_3)\,x_6 - u_4\,u_2,\ (u_1\,u_6 - u_3\,u_6 + u_5\,u_4)\,x_4 - u_1\,u_4\,u_6,\ (u_1\,u_6 - u_3\,u_6 + u_5\,u_4)\,x_3 - u_1\,u_4\,u_5,\ x_1\,(u_4 - u_6) + (-u_3 + u_5)\,x_2 + u_3\,u_6 - u_5\,u_4,\ x_6\,x_8\,(u_1\,u_2\,u_4\,u_6 - u_1\,u_2\,u_6^2 + u_1\,u_3\,u_6^2 - u_1\,u_4\,u_5\,u_6 - u_2\,u_3\,u_4\,u_6 + u_2\,u_3\,u_6^2 + u_2\,u_4^2\,u_5 - u_2\,u_4\,u_5\,u_6 - u_3^2\,u_6^2 + 2\,u_3\,u_4\,u_5\,u_6 - u_4^2\,u_5^2) + (-u_1\,u_3\,u_4\,u_6^2 + u_1\,u_4^2\,u_5\,u_6 + u_2\,u_3\,u_4\,u_6^2 - u_2\,u_4^2\,u_5\,u_6)\,x_6 + (-u_1\,u_2\,u_4^2\,u_6 + u_1\,u_2\,u_4\,u_6^2 - u_2\,u_3\,u_4\,u_6^2 + u_2\,u_4^2\,u_5\,u_6)\,x_8,\ x_2\,x_8\,(u_4\,u_2 - u_2\,u_6 + u_3\,u_6 - u_5\,u_4) + (u_1\,u_4\,u_6 - u_1\,u_6^2 - u_4\,u_2\,u_6 + u_2\,u_6^2)\,x_2 + (-u_1\,u_4\,u_6 + u_1\,u_6^2 - u_3\,u_6^2 + u_5\,u_4\,u_6)\,x_8,\ x_2\,x_6\,(u_4\,u_2 - u_2\,u_6 + u_3\,u_6 - u_5\,u_4) + (-u_2\,u_4^2 + u_4\,u_2\,u_6)\,x_2 + (-u_3\,u_4\,u_6 + u_4^2\,u_5)\,x_6] \quad \textbf{(16)}$$

```
> NormalForm(g, GenHBasis, tdeg(seq(x[j], j = 1 ..8)));
```

$$0 \quad \textbf{(17)}$$

>